

STAT 5200 Handout #11: 3-Way Factorial Design (Ch. 8)

Example: An experiment is carried out to evaluate the effects of three factors on the amount of wear sustained by fabrics in a standard abrasion test. The factors are:

- 3 • **Proportion** of filler. The fabric is “padded”, and this factor has three levels: 25%, 50%, and 75%
- 2 • **Surface** treatment. A treatment applied to the surface of the fabric. This factor has two levels, coded 1 and 2.
- 2 • **Filler** treatment. A treatment applied to the fabric filler. This factor has two levels, coded 1 and 2.

Two pieces of fabric with each combination of factor levels are made and analyzed in a random order. The response is the **WeightLoss** of the fabric in a standard abrasion test (in mg, in a standard abrasion test). There is one measurement per piece of fabric. This experiment is a 3 × 2 × 2 factorial design with 2 replicates per combination of factor levels.

```
/* STAT 5200
   factorial design (balanced)
   fabric data
   -- a 3 x 2 x 2 factorial design with 2 replicates
       per factor level combination
*/
```

```
/* Read in data */
```

```
data fabric;
  input Proportion $ Surface Filler WeightLoss @@;
  cards;
  25% 1 1 194 25% 1 1 208 25% 1 2 239 25% 1 2 187
  25% 2 1 155 25% 2 1 173
  25% 2 2 137 25% 2 2 160 50% 1 1 233 50% 1 1 241
  50% 1 2 224 50% 1 2 243
  50% 2 1 198 50% 2 1 177 50% 2 2 129 50% 2 2 98
  75% 1 1 265 75% 1 1 269
  75% 1 2 243 75% 1 2 226 75% 2 1 235 75% 2 1 229
  75% 2 2 155 75% 2 2 132
  ;
run;
```

```
/* Fit 3-way factorial model */
proc glm data=fabric plots=diagnostic;
  class Proportion Surface Filler;
  model WeightLoss = Proportion | Surface | Filler;
  title '3-way factorial model';
run;
```

include all interactions

3-way factorial model




The GLM Procedure

Dependent Variable: WeightLoss

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	11	48182.83333	4380.25758	16.30	<.0001
Error	12	3225.00000	268.75000		
Corrected Total	23	51407.83333			

R-Square	Coeff Var	Root MSE	WeightLoss Mean
0.937266	8.283080	16.39360	197.9167


Source	DF	Type III SS	Mean Square	F Value	Pr > F
Proportion	2	5967.58333	2983.79167	11.10	0.0019
Surface	1	26268.16667	26268.16667	97.74	<.0001
Proportion*Surface	2	1186.08333	593.04167	2.21	0.1527
Filler	1	6800.66667	6800.66667	25.30	0.0003
Proportion*Filler	2	3529.08333	1764.54167	6.57	0.0119
Surface*Filler	1	3952.66667	3952.66667	14.71	0.0024
Propor*Surfac*Filler	2	478.58333	239.29167	0.89	0.4360



```

proc transreg data=fabric;
  model boxcox( WeightLoss / lambda=-1 to 5 by 0.25)
    = class(Proportion | Surface | Filler );
  title1 'Check for Box-Cox Transformation';
run;

```



```

/* Look at a different way to create interaction plots */
/* Here, SLICEBY tells which variable to use as plotting
   characters,
   and PLOTBY tells which variable to separate on for
   three-way interaction plots */

```

```


proc glimmix data=fabric plots=residualpanel;
  class Proportion Surface Filler;
  model WeightLoss = Proportion | Surface | Filler;
  lsmeans Proportion*Filler /
    plot=mean(sliceby=Filler join);
  lsmeans Surface*Filler / plot=mean(sliceby=Filler join);
  lsmeans Proportion*Surface*Filler /
    plot=mean(sliceby=Surface join plotby=Filler);
  lsmeans Proportion*Surface*Filler /
    plot=mean(sliceby=Proportion join plotby=Surface)
    adjust=tukey lines;
  title1 '3-way factorial model, with interaction plots';
run;







```

3-way factorial model, with interaction plots

The GLIMMIX Procedure

Type III Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
Proportion	2	12	11.10	0.0019
Surface	1	12	97.74	<.0001
Proportion*Surface	2	12	2.21	0.1527
Filler	1	12	25.30	0.0003
Proportion*Filler	2	12	6.57	0.0119
Surface*Filler	1	12	14.71	0.0024
Propor*Surfac*Filler	2	12	0.89	0.4360





**Tukey Grouping for Propor*Surfac*Filler
Least Squares Means (Alpha=0.05)**

**LS-means with the same
letter are not significantly
different.**

Proportion of Filler	Surface Treatment	Filler Treatment	Estimate			
75%	1	1	267.00		A	
50%	1	1	237.00	B	A	
75%	1	2	234.50	B	A	
50%	1	2	233.50	B	A	
75%	2	1	232.00	B	A	
25%	1	2	213.00	B	A	C
25%	1	1	201.00	B	D	C
50%	2	1	187.50	B	D	C
25%	2	1	164.00	E	D	C
25%	2	2	148.50	E	D	C
75%	2	2	143.50	E	D	
50%	2	2	113.50	E		